
Chapter 1

The word library

Bit vector (or word)1 is one of the fundamental data objects in hardware specification
and verification. The modelling of bit vectors is a key to the success of a hardware
verification project. This library attempt to provide a general, flexible infrastructure for
reasoning about words. The description begins with a discussion of approach used by
the library to model words. This is followed by a summary of the facilities available in
the library. Chapter 2 contains the reference entries of all ML functions, and the last
chapter lists all theorems stored in the library.

1.1 Modelling bit vectors

An abstract model of words should encompass all their basic properties. It should be
independent of any concrete representation. The basic abstract properties of words are:

• a word is a vector of n elements;

• the size of a given word n is constant;

• all elements are of the same type;

• an individual element is accessed via its index.

Suppose that w is a word of size n, it can be written as

w = [|wn−1wn−2 . . . w1w0|]

where wi represents the ith bit of the word w. We adopt the convention that the bits
are indexed from the right hand side starting from 0. The index operation w[i] accesses
the ith bit of a word for all i less then n. A segment operation extracts a segment from
a word. For example,

w[m, k] = [|wk+m−1 . . . wk|] (1.1)

where (k +m) ≤ n is a m-bit segment of the word w starting from the kth bit.

1The two terms, bit vector and word will be used interchangeably in this manual.

1

A word concatenation operation • can be defined as

A•B = [|an−1 . . . a0|] • [|bm−1 . . . b0|] (1.2)

= [|an−1 . . . a0bm−1 . . . b0|]

which builds a word of size n+m from two words of size n and m, respectively.
Since words of all sizes share these basic properties, a base type of some kind would

be a starting point for modelling words. This base type should then be parameterized
with the size and the type of the elements. This suggests a dependent type of the form

: (α, n)word

where α is the type of the elements and n is the size. In the current version of the HOL
system, it is possible to define a polymorphic type :(α)word which takes the element
type as a parameter, but it is not possible to parameterize a type with natural numbers.
There is also difficulty in defining a real abstract type in the current version of HOL.

To overcome the difficulties mentioned above, the approach used in implementing the
word library uses facilities available in the current version of HOL only. First of all, it
defines a polymorphic type :(*)word to represent generic words. This allows one to use
different types to represent the bits according to the requirements of one’s applications.
For example, :(bool)word is suitable for many hardware applications using two-value
logic.

Dependent types are simulated using restricted universal quantifications. A restricted
universal quantification is written in the form

∀x :: P. t[x]

where if x : α then P can be any term of type α→ bool; this denotes the quantification
of x over those values satisfying P . The semantics of this quantification is defined by
the following equation:

`def ∀x :: P. t[x] = ∀x.P x ⊃ t[x] (1.3)

Suppose that P is a predicate PWORDLENn which returns T when applied to a word w
if and only if w is an n-bit word, then the expression

∀w :: PWORDLENn. . . .

can be read as ‘for all n-bit words w, . . .’. For a specific value of n, say 8, one can define
a predicate word8 by the definition

`def word8 = PWORDLEN 8.

This predicate can then be used in expressions, such as ∀w :: word8. . . . Since the syntax
of restricted quantification resembles the syntax of types closely and the semantics of

Figure 1.1: The ancestry of theories

the quantification is suitably defined, using this to simulate dependent types is very
comprehensible.

As we cannot define a real abstract type in HOL, the list type is used as the underlying
representation of the polymorphic type :(*)word. However, through disciplined use
of system functions and properties derived for the new type, direct reference to the
underlying representation is minimized. For example, when defining new constants,
constant specification is used to specify the abstract properties of the new constant
instead of using constant definition which needs access to the representation. In the
development of the library, the proofs of some basic theorems about words have to
refer to the underlying lists. After a small number of basic theorems are derived, one
can proceed to reason about words on a more abstract level without resorting to the
underlying representation.

1.2 The library

The word library consists of several theories and some ML functions implementing tac-
tics and conversions which manipulate words. The ancestry of the theories is illustrated
in Figure 1.1. The theories whose names begin with word contains definitions of generic
constants and theorems asserting general properties of words. These generic constants
are polymorphic and can be applied to words of any types. There are three such theories
in the library, namely word base, word bitop and word num. As boolean words are used
most often, the theories whose names begin with bword are about this type of words.
The subsections below describe individual theories in more detail.

PWORDLEN:num -> ((*)word -> bool)

PWORDLENnw = T iff w is an n-bit word

WORDLEN:(*)word -> num

WORDLENw = n

BIT :num -> ((*)word -> *)

BIT i [|an−1 . . . ai . . . a0|] = ai

WSEG:num -> (num -> ((*)word -> (*)word))

WSEGmk [|an−1 . . . ak+m−1 . . . ak . . . a0|] = [|ak+m−1 . . . ak|]

WCAT:(*)word # (*)word -> (*)word

WCAT([|an−1 . . . a0|] , [|bm−1 . . . b0|]) = [|an−1 . . . a0bm−1 . . . b0|]

Table 1.1: Basic constants in the theory word base

1.2.1 The basics: the theory word base

First of all, the polymorphic type :(*)word is defined in this theory. It is defined using
define_type with the following specification:

‘word = WORD (*)list‘

The basic constants denoting the functions of indexing, segmenting and concatenation
of words described in Section 1.1 are BIT, WSEG and WCAT, respectively. The predi-
cate PWORDLEN for discriminating the size of words and a function named WORDLEN

returning the size of a word are also defined in this theory. The types and specifications
of these constants are listed in Table 1.1. Several constants denoting some simple func-
tions on words are also defined for convenient, such as MSB for most significant bit.
These are listed in Table 1.2.

A number of theorems stating the properties of the basic constants are stored in
this theory. Some of the more important ones are discussed below. The theorem
WSEG_PWORDLEN states that the size of the word resulting from taking an m-bit segment
from an n-bit word is m providing that k +m ≤ n where k is the starting bit.

HOL Theorem (WSEG_PWORDLEN)

` ∀n. ∀w :: PWORDLENn.

∀mk.m+ k ≤ n ⊃ PWORDLENm (WSEGmkw)

A nested WSEG expression can be simplified providing that the sizes and starting bits
satisfy certain conditions. This is asserted by the theorem WSEG_WSEG.

LSB :(*)word -> *

` ∀n. ∀w :: PWORDLENn. 0 < n ⊃ (LSBw = BIT 0w)

MSB :(*)word -> *

` ∀n. ∀w :: PWORDLENn.

0 < n ⊃ (MSBw = BIT (PREn)w)

WSPLIT:num -> ((*)word -> (*)word # (*)word)

` (∀n. ∀w :: PWORDLENn.

∀m.m ≤ n ⊃ (WCAT (WSPLITmw) = w)) ∧
(∀nm. ∀w1 :: PWORDLENn. ∀22 :: PWORDLENm.

WSPLITm (WCAT (w1, w2)) = w1, w2))

` ∀n. ∀w :: PWORDLENn.

∀k. k ≤ n ⊃ (WSPLIT k w = WSEG (n− k) k w,WSEG k 0w)

Table 1.2: Other constants in the theory word base

HOL Theorem (WSEG_WSEG)

` ∀n. ∀w :: PWORDLENn.

∀m1 k1 m2 k2.m1 + k1 ≤ n ∧m2 + k2 ≤ m1 ⊃
(WSEGm2 k2 (WSEGm1 k1 w) = WSEGm2 (k1 + k2)w)

The theorem WCAT_PWORDLEN states that the size of the result of the word concatenation
operation is the sum of the sizes of its operands.

HOL Theorem (WCAT_PWORDLEN)

` ∀n1.∀w1 :: PWORDLENn1.

∀n2.∀w2 :: PWORDLENn2.

PWORDLEN (n1 + n2) (WCAT (w1, w2))

The associativity of the WCAT operation is asserted by the theorem WCAT_ASSOC.

HOL Theorem (WCAT_ASSOC)

` ∀w1 w2 w3.

WCAT (w1,WCAT (w2, w3)) = WCAT (WCAT (w1, w2), w3)

The theorem WSEG_WCAT_WSEG asserts that taking a segment from a word which is built
by concatenating two words w1 and w2 is equivalent to taking the appropriate segments
from each word and then concatenating them provided that the segment spans across
the boundary of the two words.

HOL Theorem (WSEG_WCAT_WSEG)

` ∀n1 ∀n2.∀w1 :: PWORDLENn1.∀w2 :: PWORDLENn2.

∀mk.

m+ k ≤ n1 + n2 ∧ k < n2 ∧ n2 ≤ m+ k ⊃
(WSEGmk (WCAT (w1, w2)) =

WCAT (WSEG ((m+ k)− n2) 0w1,WSEG (n2 − k) k w2))

1.2.2 Generic bitwise operators: the theory word bitop

Definitions in this theory include predicates for bitwise operators, predicates on proper-
ties of bits and shift operators.

Two predicates, PBITOP and PBITBOP are defined for quantifying bitwise operators.
When applied to a suitably typed word function op, they will return T if and only if op
is a bitwise unary or binary operator, respectively. The meaning of bitwise is that the
operator preserves the size and the operation on each bit is independent of other bits.
Note that as these predicates are polymorphic the type of the bits can be anything. The
exact definitions of these predicates are as follows:

PBITOP:((*)word -> (**)word) -> bool

PBITOP op = T iff op is a bitwise unary operator

`def ∀op.PBITOP op =

(∀n. ∀w :: PWORDLENn.

PWORDLENn (opw) ∧
(∀mk.m+ k ≤ n ⊃ (op (WSEGmkw) = WSEGmk (opw)))

PBITBOP:((*)word -> (**)word -> (***)word) -> bool

PBITBOP op = T iff op is a bitwise binary operator

`def ∀op.PBITBOP op =

(∀n. ∀w1 :: PWORDLENn. ∀w2PWORDLENn.

PWORDLENn (opw1 w2) ∧
(∀mk.m+ k ≤ n ⊃

(op (WSEGmkw1) (WSEGmkw2) = WSEGmk (opw1 w2)))

SHR :bool -> * -> (*)word -> ((*)word # *)

SHR f b [|an−1 . . . a1a0|] =

{
([|an−1an−1 . . . a1|] , a0) if f = T
([|b an−1 . . . a1|] , a0) if f = F

SHL :bool -> (*)word -> * -> (* # (*)word)

SHL f [|an−1an−2 . . . a0|] b =

{
(an−1, [|an−2 . . . a0a0|]) if f = T
(an−1, [|an−2 . . . a0 b|]) if f = F

Table 1.3: Shift operators

Two higher-order functions, FORALLBITS and EXISTSABIT, are defined for testing
whether the bits of a word have certain properties. The term FORALLBITSP w eval-
uates to T if and only if all the bits in the word w satisfy the predicate P . The term
EXISTSABITP w evaluates to T if and only if there exists one or more bits in the word
w satisfying the predicate P . The higher-order function WMAP defined in this theory is
analogous to the function MAP on lists. The meaning of the expression WMAP f w is to
apply the function f to each bit of the word w.

Also in this theory are the definitions of two generic shift operators: SHL and SHR.
Their types and specification is listed in Table 1.3. Both take three arguments and
return a pair. The first argument is a boolean value indicating the kind of operation
to be performed. The second and the third arguments to SHR are a single bit and a
word, respectively. The order of these two arguments to SHL is reversed. Depending on
the value of the boolean and single bit argument, these operators can perform either a
logical shift, an arithmetic shift or a rotation operation. If the boolean argument is T,
the single bit argument is not used. SHR shifts its operand one bit to the right and the
left-most bit is duplicated to fill the vacant position, thus, implementing an arithmetic
shift. If the boolean argument is F, SHR fills the vacant position with the single bit
argument. If this bit is the right-most bit of the operand, a rotation is performed. If it
has value 0, it results in a logical shift. The operation performed by SHL is similar. The
pair returned by these operators consists of a word which is the operation result and a
single bit which is the bit shifted out of the operand.

A number of theorems asserting the operational behaviour of these operators and
their relationship with the basic constants WCAT and WSEG are stored in this theory.
The theorems SHR_WSEG and SHL_WSEG state the equivalence between a shift expression
and a combination of WCAT and WSEG. Thus, an expression involving shift operators
can be simplified to one which only involves the basic word operations.

HOL Theorem (SHR_WSEG)

` ∀n. ∀w :: PWORDLENn.

∀mk.m+ k ≤ n ⊃ 0 < m ⊃
(∀f b. SHR f b (WSEGmkw) =

(f ⇒ WCAT (WSEG 1 (k + (m− 1))w,WSEG (m− 1) (k + 1)w) |
WCAT (WORD [b],WSEG (m− 1) (k + 1)w)),

BIT k w)

HOL Theorem (SHL_WSEG)

` ∀n. ∀w :: PWORDLENn.

∀mk.m+ k ≤ n ⊃ 0 < m ⊃
(∀f b. SHL f (WSEGmkw) b =

BIT (k + (m− 1))w,

(f ⇒ WCAT (WSEG (m− 1) k w,WSEG 1 k w) |
WCAT (WSEG (m− 1) k w,WORD [b])))

1.2.3 Boolean bitwise operators: the theory bword bitop

In this theory, a small set of boolean bitwise operators are defined and theorems assert-
ing that they are bitwise operators are proved. The boolean bitwise operators are:

WNOT :bool word -> bool word bitwise negation
WAND :bool word -> bool word -> bool word bitwise AND
WOR :bool word -> bool word -> bool word bitwise OR
WXOR :bool word -> bool word -> bool word bitwise exclusive-OR

The theorems stating that they are bitwise are:

PBITOP_WNOT ` PBITOP WNOT
PBITBOP_WAND ` PBITBOP WAND
PBITBOP_WOR ` PBITBOP WOR
PBITBOP_WXOR ` PBITBOP WXOR

1.2.4 Natural numbers and words: the theory word num

Words are often interpreted as natural numbers. In this theory, two constants are de-
fined to map generic words to natural numbers and vice versa:

NVAL:(* -> num) -> num -> (*)word -> num

NVAL f b w returns the numeric value of w. f is a function mapping a bit to its
numeric value and b is the base or radix of the word.

NWORD:num -> (num -> *) -> num -> num -> (*)word

NWORD n f ′ b m returns an n-bit word representing the value of m. f ′ is a
function mapping a number to a bit and b is the base.

The upper bound of the numeric value of a word is stated by the theorem NVAL_MAX.

HOL Theorem (NVAL_MAX)

` ∀f b. (∀x. f x < b) ⊃
∀n. ∀w :: PWORDLENn.NVAL f bw < (b EXP n)

Provided that the bit value function f satisfies ∀x. f x < b, the numeric value of a word
w is always less than bn. The theorem NVAL_WCAT states that the value of a word can be
calculated from the values of its segments.

HOL Theorem (NVAL_WCAT)

` ∀nm. ∀w1 :: PWORDLENn.

∀w2 :: PWORDLENm.

∀f b.
NVAL f b (WCAT (w1, w2)) =

(NVAL f bw1 × (b EXP m)) + (NVAL f bw2)

The theorem stating the size of the result of mapping from natural number to word is
NWORD_PWORDLEN.

HOL Theorem (NWORD_PWORDLEN)

` ∀n f bm.PWORDLENn (NWORDn f bm)

1.2.5 Boolean words and numbers: the theory bword num

In this theory, two functions mapping between a single bit and number are defined first.
Then, the constants denoting the mapping between boolean words and natural numbers
are defined in terms of these bit mapping functions and the generic word–num mapping
functions described in Section 1.2.4.

BV :bool -> num

` ∀b.BV b = (b⇒ SUC 0 | 0)

VB :num -> bool

` ∀n.VBn = ¬((nMOD 2) = 0)



10 Chapter 1. The word library

BNVAL:bool word -> num

BNVAL w returns the numeric value of w. `def BNVAL w = NVAL BV 2 w

NBWORD:num -> num -> bool word

NBWORD n m returns a n-bit word representing the value of m.
`def NBWORD n m = NWORD n VB 2 m

The functions BNVAL and NBWORD are inverse to each other in the set of numbers
less than 2n where n is the size of the word. The following theorems state the basic
properties of these mapping functions.

HOL Theorem (VB_BV)

` ∀x.VB (BVx) = x

HOL Theorem (BV_VB)

` ∀x. x < 2 ⊃ (BV (VBx) = x)

HOL Theorem (NBWORD_BNVAL)

` ∀n. ∀w :: PWORDLENn.NBWORDn (BNVALw) = w

HOL Theorem (BNVAL_NBWORD)

` ∀nm.
m < (2 EXP n) ⊃ (BNVAL (NBWORDnm) = m)

HOL Theorem (PWORDLEN_NBWORD)

` ∀nm.PWORDLENn (NBWORDnm)

HOL Theorem (NBWORD_MOD)

` ∀nm.NBWORDn (m MOD (2 EXP n)) = NBWORDnm

The theorem NBWORD_SUC asserts the fact that converting a number m to a word can be
performed bit by bit recursively.

HOL Theorem (NBWORD_SUC)

` ∀nm.NBWORD (SUCn)m =

WCAT (NBWORDn (m DIV 2),WORD [VB (m MOD 2)])

The theorem WSEG_NBWORD states that taking an m-bit segment of an n-bit word mapped
to by NBWORD from a number l is equivalent to mapping the quotient of l divided by
2k to an m-bit word.

HOL Theorem (WSEG_NBWORD)

` ∀mk n.m+ k ≤ n ⊃
(∀l.WSEGmk (NBWORDn l) = NBWORDm (l DIV (2 EXP k)))

1.2.6 Boolean word arithmetic: the theory bword arith

This theory is about addition of boolean words. Two methods of computing the carry
value of each bit are defined: ACARRY uses addition and ICARRY uses logical operations
∧ and ∨. The theorem asserting the equivalence of these methods is ACARRY_EQ_ICARRY.

The theorem ADD_WORD_SPLIT states that addition of two words can be carried out in
segments.

HOL Theorem (ADD_WORD_SPLIT)

` ∀n1 n2.∀w1 w2 :: PWORDLEN (n1 + n2).∀cin.
NBWORD (n1 + n2) (BNVALw1 + BNVALw2 + BV cin) =

WCAT (NBWORDn1 (BNVAL (WSEGn1 n2 w1) + BNVAL (WSEGn1 n2 w2) +

BV (ACARRYn2 w1 w2 cin)),

NBWORDn2 (BNVAL (WSEGn2 0w1) + BNVAL (WSEGn2 0w2) +

BV cin))

The theorem WSEG_NBWORD_ADD asserts that taking a segment of the sum of two words
is equal to taking the corresponding segments of the words then summing them up.

HOL Theorem (WSEG_NBWORD_ADD)

` ∀n. ∀w1 w2 :: PWORDLENn. ∀mk cin.m+ k ≤ n ⊃
(WSEGmk (NBWORDn (BNVALw1 + BNVALw2 + BV cin)) =

NBWORDm (BNVAL (WSEGmkw1) + BNVAL (WSEGmkw2) +

BV (ACARRY k w1 w2 cin)))

1.2.7 Proof tools

The word library currently has a small set of tools in the form of conversions and tactics
for manipulating words. These include the following:

BIT CONV :conv When applied to a term as the left hand side of the following theorem,
this conversion returns the theorem

` BIT k (WORD[wn−1; . . . ;wk; . . . ;w0]) = wk

WSEG CONV :conv When applied to a term as the left hand side of the following theorem,
this conversion returns the theorem

` WSEGmk (WORD[wn−1; . . . ;wk; . . . ;w0]) = [wm+k−1; . . . ;wk]

WSEG WSEG CONV :(term -> conv) When applied to a term as the left hand side of the
following theorem, the conversion WSEG_WSEG_CONV "n" returns the theorem

PWORDLENnw ` WSEGm2 k2(WSEGm1 k1w) = WSEGm2 k w

where k = k1 + k2 and n, k1, k2, m1 and m2 are numeric constants and satisfy the
following relations: k1 +m1 ≤ n and k2 +m2 ≤ m1.

PWORDLEN CONV :(term list -> conv) When applied to the term PWORDLENmtm,
the conversion PWORDLEN_CONV tms returns a theorem asserting the size of the
word tm. This theorem is in the form

A ` PWORDLENmtm = T

where the exact form of A, tm and the term list argument tms is given in the table
below:

tm tms theorem
WORD[bn−1; . . . ; b0] [ ] ` PWORDLENn (WORD[bn−1; . . . ; b0])
WSEGmk tm′ [ "n" ] PWORDLENn tm′

` PWORDLENm (WSEGmk tm′)
WCAT(tm′, tm′′) ["n1";"n2"] PWORDLENn1 tm

′,PWORDLENn2 tm
′′

` PWORDLENn (WCAT(tm′, tm′′))
where n = n1 + n2

WNOT tm′ [ ] PWORDLENn tm′

` PWORDLENn (WNOT tm′)
WAND tm′ tm′′ [ ] PWORDLENn tm′,PWORDLENn tm′′

` PWORDLENn (WAND tm′ tm′′)
WOR tm′ tm′′ [ ] PWORDLENn tm′,PWORDLENn tm′′

` PWORDLENn (WOR tm′ tm′′)
WXOR tm′ tm′′ [ ] PWORDLENn tm′,PWORDLENn tm′′

` PWORDLENn (WXOR tm′ tm′′)

PWORDLEN bitop CONV :conv When applied to a term PWORDLENn tm where tm in-
volves only bitwise operators and variables, this conversion returns the theorem

. . . ,PWORDLENnwi, . . . ` PWORDLENn tm = T

where there is one assumption PWORDLENnwi for each simple variable wi in
tm. This conversion automatically descends into the subterms until it reaches all
variables.

PWORDLEN TAC :(term list -> tactic) When applied to a goal of the form PWORDLEN

n tm, the tactic PWORDLEN_TAC tms solves it if the conversion PWORDLEN_CONV tms

returns a theorem without assumptions. Otherwise, the assumptions of the theo-
rem returned by the conversion become the new subgoals.

1.3 Working with words

The basic technique for reasoning about words with the word library is by structural
induction on the size of the word. Since the structure of words is linear and symmetric,
structural induction can be carried out from either end using the WCAT operation as the
basic constructor. In addition, structural analysis can be done at any position of a word.
In general, there are three theorems associated with each basic word function: one for
each kind of structural analysis. Considering the function NBWORD as an example,
the theorem NBWORD_SUC described in Section 1.2.5 is for structural induction from the
right hand end. The theorem NBWORD_SUC_LEFT shown below is for structural induction
from the left hand end, and the theorem NBWORD_SPLIT is for structural analysis at any
position.

HOL Theorem (NBWORD_SUC_LEFT)

` ∀nm.NBWORD (SUCn)m =

WCAT (WORD [VB ((m DIV (2) EXP (n)) MOD 2)],NBWORDnm)

HOL Theorem (NBWORD_SPLIT)

` ∀n1 n2 m.NBWORD (n1 + n2)m =

WCAT (NBWORDn1 (m DIV (2) EXP (n2)),NBWORDn2 m)

The following example uses structural induction from the right hand end to prove
a theorem about taking an n-bit segment of an (n + 1)-bit word which is the result of
converting a natural number using the function NBWORD. We first set up the goal

? – ∀nm.WSEGn 0(NBWORD(SUCn)m) = NBWORDnm.

Then, the induction tactic INDUCT_TAC is applied to the size of the word. This generates
two subgoals. The first subgoal, corresponding to the base case of the induction, is

? – WSEG 0 0(NBWORD(SUC 0)m) = NBWORD 0m.

This is trivial to solve since a zero-bit segment of a word is WORD[ ] and converting
a number to a zero-bit word always gives the same result. The second subgoal corre-
sponding to the step case of the induction is

? – ∀m.WSEG (SUCn) 0 (NBWORD (SUC (SUCn))m) = NBWORD (SUCn)m

The right hand end induction theorem for NBWORD, NBWORD_SUC, can now be used to
rewrite the goal. Rewriting the resulting goal further with the theorem WSEG_WCAT_WSEG

and simplifying the result reduces it to

? – WSEGn 0 (NBWORD (SUCn) (m DIV 2)) = NBWORDn (m DIV 2).

let WSEG_NBWORD_SUC = PROVE(
"!n m. (WSEG n 0(NBWORD (SUC n) m) = NBWORD n m)",
INDUCT_TAC THENL[
REWRITE_TAC[NBWORD0;WSEG0];
GEN_TAC THEN PURE_ONCE_REWRITE_TAC[NBWORD_SUC]
THEN RESQ_REWRITE1_TAC (SPECL["SUC n"; "1"] WSEG_WCAT_WSEG) THENL[
MATCH_ACCEPT_TAC PWORDLEN_NBWORD;
MATCH_ACCEPT_TAC PWORDLEN1;
PURE_ONCE_REWRITE_TAC[GSYM ADD1] THEN PURE_ONCE_REWRITE_TAC[ADD_0]
THEN MATCH_ACCEPT_TAC LESS_EQ_SUC_REFL;
CONV_TAC (RAND_CONV num_CONV) THEN MATCH_ACCEPT_TAC LESS_0;
CONV_TAC ((RATOR_CONV o RAND_CONV) num_CONV)
THEN PURE_REWRITE_TAC[ADD_0;LESS_EQ_MONO]
THEN MATCH_ACCEPT_TAC ZERO_LESS_EQ;
PURE_REWRITE_TAC[SUB_0;ADD_0;SUC_SUB1]
THEN PURE_ONCE_ASM_REWRITE_TAC[]
THEN RESQ_REWRITE1_TAC (SPEC "1" WSEG_WORD_LENGTH)
THEN REFL_TAC]]);;

Figure 1.2: A proof of the theorem WSEG NBWORD SUC

The induction hypothesis can then be used to solve the goal. However, as the the-
orem WSEG_WCAT_WSEG is restricted universally quantified, ordinary rewriting tactics,
such as REWRITE_TAC, cannot use it to rewrite the goal. Special tactics are required. The
res quan library provides the basic facilities for manipulating restricted quantifications[1].
The complete proof is listed in Figure 1.2.



Chapter 2

ML Functions in the word Library

This chapter provides documentation on all the ML functions that are made available in
HOL when the word library is loaded. This documentation is also available online via
the help facility.

BIT_CONV

BIT_CONV : conv

Synopsis
Computes by inference the result of accessing a bit in a word.

Description
For any word of the form WORD[b(n-1);...;bk;...;b0], the result of evaluating

BIT_CONV "BIT k (WORD [b(n-1);...;bk;...;b0])"

is the theorem

|- BIT k (WORD [b(n-1);...;bk;...;b0]) = bk

The bits are indexed form the end of the list and starts from 0.

Failure
BIT_CONV tm fails if tm is not of the form "BIT k w" where w is as described above, or k

is not less than the size of the word.

See also
WSEG_CONV

PWORDLEN_bitop_CONV

PWORDLEN_bitop_CONV : conv

15

Synopsis
Computes by inference the predicate asserting the size of a word.

Description
For a term tm of type :(bool)word involving only a combination of bitwise operators
WNOT, WAND, WOR, WXOR and variables, the result of evaluating

PWORDLEN_bitop_CONV "PWORDLEN n tm"

is the theorem

..., PWORDLEN n vi, ... |- PWORDLEN n tm = T

Each free variable occurred in tm will have a corresponding clause in the assumption.
This conversion recursively descends into the subterms of tm until it reaches all simple
variables.

Failure
PWORDLEN_bitop_CONV tm fails if constants other than those mentioned above occur in
tm.

See also
PWORDLEN_CONV, PWORDLEN_TAC

PWORDLEN_CONV

PWORDLEN_CONV : term list -> conv

Synopsis
Computes by inference the predicate asserting the size of a word.

Description
For any term tm of type :(*)word, the result of evaluating

PWORDLEN_CONV tms "PWORDLEN n tm"

where n must be a numeric constant, is the theorem

A |- PWORDLEN n tm = T

where the new assumption(s) A depends on the actual form of the term tm.

If tm is an application of the unary bitwise operator WNOT, i.e., tm = WNOT tm’, then
A will be PWORDLEN n tm’. If tm is an application of one of the binary bitwise oper-
ators: WAND, WOR and WXOR, then A will be PWORDLEN n tm’, PWORDLEN n tm’’. If tm is
WORD [b(n-1);...;b0], then A is empty. The length of the list must agree with n. In all
above cases, the term list argument is irrelevant. An empty list could be supplied.

If tm is WSEG n k tm’, then the term list tms should be [N] which indicates the size of
tm’, and the assumption A will be PWORDLEN N tm’.

If tm is WCAT(tm’,tm’’), then the term list tms should be [n1;n2] which tells the sizes of
the words to be concatenated. The assumption will be PWORDLEN n1 tm’, PWORDLEN n2 tm’’.
The value of n must be the sum of n1 and n2.

Failure
PWORDLEN_CONV tms tm fails if tm is not of the form described above.

See also
PWORDLEN_bitop_CONV, PWORDLEN_TAC

PWORDLEN_TAC

PWORDLEN_TAC : term list -> tactic

Synopsis
Tactic to solve a goal about the size of a word.

Description
When applied to a goal A ?- PWORDLEN n tm, the tactic PWORDLEN_TAC tms solves it if the
conversion PWORDLEN_CONV tms returns a theorem

A’ |- PWORDLEN n tm

where A’ is either empty or every clause in it occurs in the assumption of the goal A.
Otherwise, each clause in A’ which does not appear in A becomes a new subgoal.

Failure
PWORDLEN_TAC tms fails if the corresponding conversion PWORDLEN_CONV fails.

See also
PWORDLEN_CONV

WSEG_CONV

WSEG_CONV : conv

Synopsis
Computes by inference the result of taking a segment from a word.

Description
For any word of the form WORD[b(n-1);...;bk;...;b0], the result of evaluating

WSEG_CONV "WSEG m k (WORD [b(n-1);...;bk;...;b0])",

where m and k must be numeric constants, is the theorem

|- WSEG m k (WORD [b(n-1);...;bk;...;b0]) = [b(m+k-1);...;bk]

The bits are indexed form the end of the list and starts from 0.

Failure
WSEG_CONV tm fails if tm is not of the form described above, or m + k is not less than the
size of the word.

See also
BIT_CONV, WSEG_WSEG_CONV

WSEG_WSEG_CONV

WSEG_WSEG_CONV : term -> conv

Synopsis
Computes by inference the result of taking a segment from a segment of a word.

Description
For any word w of size n, the result of evaluating

WSEG_WSEG_CONV "n" "WSEG m2 k2 (WSEG m1 k1 w)"

where m2, k2, m1 and k1 must be numeric constants, is the theorem

PWORDLEN n w |- WSEG m2 k2 (WSEG m1 k1 w) = WSEG m2 k w

where k is a numeric constant whose value is the sum of k1 and k2.

Failure
WSEG_WSEG_CONV tm fails if tm is not of the form described above, or the relations k1 + m1 <= n

and k2 + m2 <= m1 are not satisfied.

See also
BIT_CONV, WSEG_CONV

Chapter 3

Pre-proved Theorems

The sections that follow list all theorems in the theory word. The theorems listed in this
chapter will be available by name at the top-level when the theories in which they are
declared are open-ed.

3.1 The theory word base

BIT0 (word base)
|- !b. BIT 0 (WORD [b]) = b

BIT_DEF (word base)
|- !k l. BIT k (WORD l) = ELL k l

BIT_EQ_IMP_WORD_EQ (word base)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n).

(!k. k < n ==> (BIT k w1 = BIT k w2)) ==> (w1 = w2)

BIT_WCAT1 (word base)
|- !n (w::PWORDLEN n) b. BIT n (WCAT (WORD [b],w)) = b

BIT_WCAT_FST (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2) k.

n2 <= k /\ k < n1 + n2 ==> (BIT k (WCAT (w1,w2)) = BIT (k - n2) w1)

BIT_WCAT_SND (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2) k.

k < n2 ==> (BIT k (WCAT (w1,w2)) = BIT k w2)

BIT_WSEG (word base)
|- !n (w::PWORDLEN n) m k j.

m + k <= n ==> j < m ==> (BIT j (WSEG m k w) = BIT (j + k) w)

ii_internalword_base0_def (word base)
|- ii_internalword_base0 =

(\a. ii_internal_mk_word ((\a. CONSTR 0 a (\n. BOTTOM)) a))

21



22 Chapter 3. Pre-proved Theorems

LSB (word base)
|- !n (w::PWORDLEN n). 0 < n ==> (LSB w = BIT 0 w)

LSB_DEF (word base)
|- !l. LSB (WORD l) = LAST l

MSB (word base)
|- !n (w::PWORDLEN n). 0 < n ==> (MSB w = BIT (PRE n) w)

MSB_DEF (word base)
|- !l. MSB (WORD l) = HD l

PWORDLEN (word base)
|- !n w. PWORDLEN n w = (WORDLEN w = n)

PWORDLEN0 (word base)
|- !w. PWORDLEN 0 w ==> (w = WORD [])

PWORDLEN1 (word base)
|- !x. PWORDLEN 1 (WORD [x])

PWORDLEN_DEF (word base)
|- !n l. PWORDLEN n (WORD l) = (n = LENGTH l)

WCAT0 (word base)
|- !w. (WCAT (WORD [],w) = w) /\ (WCAT (w,WORD []) = w)

WCAT_11 (word base)
|- !m n (wm1::PWORDLEN m) (wm2::PWORDLEN m) (wn1::PWORDLEN n)

(wn2::PWORDLEN n).
(WCAT (wm1,wn1) = WCAT (wm2,wn2)) = (wm1 = wm2) /\ (wn1 = wn2)

WCAT_ASSOC (word base)
|- !w1 w2 w3. WCAT (w1,WCAT (w2,w3)) = WCAT (WCAT (w1,w2),w3)

WCAT_DEF (word base)
|- !l1 l2. WCAT (WORD l1,WORD l2) = WORD (APPEND l1 l2)

WCAT_PWORDLEN (word base)
|- !n1 (w1::PWORDLEN n1) n2 (w2::PWORDLEN n2).

PWORDLEN (n1 + n2) (WCAT (w1,w2))

WCAT_WSEG_WSEG (word base)
|- !n (w::PWORDLEN n) m1 m2 k.

m1 + (m2 + k) <= n ==>
(WCAT (WSEG m2 (m1 + k) w,WSEG m1 k w) = WSEG (m1 + m2) k w)

WORD (word base)
|- WORD = ii_internalword_base0

WORDLEN_DEF (word base)
|- !l. WORDLEN (WORD l) = LENGTH l

WORDLEN_SUC_WCAT (word base)
|- !n w.

PWORDLEN (SUC n) w ==>
?(b::PWORDLEN 1) (w’::PWORDLEN n). w = WCAT (b,w’)

WORDLEN_SUC_WCAT_BIT_WSEG (word base)
|- !n (w::PWORDLEN (SUC n)). w = WCAT (WORD [BIT n w],WSEG n 0 w)

WORDLEN_SUC_WCAT_BIT_WSEG_RIGHT (word base)
|- !n (w::PWORDLEN (SUC n)). w = WCAT (WSEG n 1 w,WORD [BIT 0 w])

WORDLEN_SUC_WCAT_WSEG_WSEG (word base)
|- !w::PWORDLEN (SUC n). w = WCAT (WSEG 1 n w,WSEG n 0 w)

WORDLEN_SUC_WCAT_WSEG_WSEG_RIGHT (word base)
|- !w::PWORDLEN (SUC n). w = WCAT (WSEG n 1 w,WSEG 1 0 w)

WORD_11 (word base)
|- !l l’. (WORD l = WORD l’) = (l = l’)

word_11 (word base)
|- !a a’. (WORD a = WORD a’) = (a = a’)

word_Ax (word base)
|- !f. ?fn. !a. fn (WORD a) = f a

word_Axiom (word base)
|- !f. ?fn. !a. fn (WORD a) = f a

word_cases (word base)
|- !w. ?l. w = WORD l

word_case_cong (word base)
|- !f’ f M’ M.

(M = M’) /\ (!a. (M’ = WORD a) ==> (f a = f’ a)) ==>
(word_case f M = word_case f’ M’)

word_case_def (word base)
|- !f a. word_case f (WORD a) = f a

WORD_CONS_WCAT (word base)
|- !x l. WORD (x::l) = WCAT (WORD [x],WORD l)

WORD_DEF (word base)
|- !l. WORD l = ABS_word (Node l [])

word_induct (word base)
|- !P. (!l. P (WORD l)) ==> !w. P w

word_induction (word base)
|- !P. (!l. P (WORD l)) ==> !w. P w

word_ISO_DEF (word base)
|- (!a. ABS_word (REP_word a) = a) /\

!r.
TRP (\v tl. (?l. v = l) /\ (LENGTH tl = 0)) r =
(REP_word (ABS_word r) = r)

word_nchotomy (word base)
|- !w. ?l. w = WORD l

WORD_PARTITION (word base)
|- (!n (w::PWORDLEN n) m. m <= n ==> (WCAT (WSPLIT m w) = w)) /\

!n m (w1::PWORDLEN n) (w2::PWORDLEN m).
WSPLIT m (WCAT (w1,w2)) = (w1,w2)

word_repfns (word base)
|- (!a. ii_internal_mk_word (ii_internal_dest_word a) = a) /\

!r.
(\a0.

!’word’.
(!a0.

(?a. a0 = (\a. CONSTR 0 a (\n. BOTTOM)) a) ==>
’word’ a0) ==>

’word’ a0) r =
(ii_internal_dest_word (ii_internal_mk_word r) = r)

word_size_def (word base)
|- !f a. word_size f (WORD a) = 1 + list_size f a

word_size_full_def (word base)
|- !f a. word_size f (WORD a) = 1 + list_size f a

WORD_SNOC_WCAT (word base)
|- !l x. WORD (SNOC x l) = WCAT (WORD l,WORD [x])

WORD_SPLIT (word base)
|- !n1 n2 (w::PWORDLEN (n1 + n2)). w = WCAT (WSEG n1 n2 w,WSEG n2 0 w)

word_TY_DEF (word base)
|- ?rep.

TYPE_DEFINITION
(\a0.

!’word’.
(!a0.

(?a. a0 = (\a. CONSTR 0 a (\n. BOTTOM)) a) ==>
’word’ a0) ==>

’word’ a0) rep

WSEG0 (word base)
|- !k w. WSEG 0 k w = WORD []

WSEG_BIT (word base)
|- !n (w::PWORDLEN n) k. k < n ==> (WSEG 1 k w = WORD [BIT k w])

WSEG_DEF (word base)
|- !m k l. WSEG m k (WORD l) = WORD (LASTN m (BUTLASTN k l))

WSEG_PWORDLEN (word base)
|- !n (w::PWORDLEN n) m k. m + k <= n ==> PWORDLEN m (WSEG m k w)

WSEG_WCAT1 (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2).

WSEG n1 n2 (WCAT (w1,w2)) = w1

WSEG_WCAT2 (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2).

WSEG n2 0 (WCAT (w1,w2)) = w2

WSEG_WCAT_WSEG (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2) m k.

m + k <= n1 + n2 /\ k < n2 /\ n2 <= m + k ==>
(WSEG m k (WCAT (w1,w2)) =
WCAT (WSEG (m + k - n2) 0 w1,WSEG (n2 - k) k w2))

WSEG_WCAT_WSEG1 (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2) m k.

m <= n1 /\ n2 <= k ==>
(WSEG m k (WCAT (w1,w2)) = WSEG m (k - n2) w1)

WSEG_WCAT_WSEG2 (word base)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2) m k.

m + k <= n2 ==> (WSEG m k (WCAT (w1,w2)) = WSEG m k w2)

WSEG_WORDLEN (word base)
|- !n (w::PWORDLEN n) m k. m + k <= n ==> (WORDLEN (WSEG m k w) = m)

WSEG_WORD_LENGTH (word base)
|- !n (w::PWORDLEN n). WSEG n 0 w = w

WSEG_WSEG (word base)
|- !n (w::PWORDLEN n) m1 k1 m2 k2.

m1 + k1 <= n /\ m2 + k2 <= m1 ==>
(WSEG m2 k2 (WSEG m1 k1 w) = WSEG m2 (k1 + k2) w)

WSPLIT_DEF (word base)
|- !m l. WSPLIT m (WORD l) = (WORD (BUTLASTN m l),WORD (LASTN m l))

WSPLIT_PWORDLEN (word base)
|- !n (w::PWORDLEN n) m.

m <= n ==>
PWORDLEN (n - m) (FST (WSPLIT m w)) /\
PWORDLEN m (SND (WSPLIT m w))

WSPLIT_WSEG (word base)
|- !n (w::PWORDLEN n) k.

k <= n ==> (WSPLIT k w = (WSEG (n - k) k w,WSEG k 0 w))

WSPLIT_WSEG1 (word base)
|- !n (w::PWORDLEN n) k.

k <= n ==> (FST (WSPLIT k w) = WSEG (n - k) k w)

WSPLIT_WSEG2 (word base)
|- !n (w::PWORDLEN n) k. k <= n ==> (SND (WSPLIT k w) = WSEG k 0 w)

3.2 The theory word bitop

EXISTSABIT (word bitop)
|- !n (w::PWORDLEN n) P. EXISTSABIT P w = ?k. k < n /\ P (BIT k w)

EXISTSABIT_DEF (word bitop)
|- !P l. EXISTSABIT P (WORD l) = SOME_EL P l

EXISTSABIT_WCAT (word bitop)
|- !w1 w2 P.

EXISTSABIT P (WCAT (w1,w2)) = EXISTSABIT P w1 \/ EXISTSABIT P w2

EXISTSABIT_WSEG (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==> !P. EXISTSABIT P (WSEG m k w) ==> EXISTSABIT P w

FORALLBITS (word bitop)
|- !n (w::PWORDLEN n) P. FORALLBITS P w = !k. k < n ==> P (BIT k w)

FORALLBITS_DEF (word bitop)
|- !P l. FORALLBITS P (WORD l) = ALL_EL P l

FORALLBITS_WCAT (word bitop)
|- !w1 w2 P.

FORALLBITS P (WCAT (w1,w2)) = FORALLBITS P w1 /\ FORALLBITS P w2

FORALLBITS_WSEG (word bitop)
|- !n (w::PWORDLEN n) P.

FORALLBITS P w ==> !m k. m + k <= n ==> FORALLBITS P (WSEG m k w)

NOT_EXISTSABIT (word bitop)
|- !P w. ~EXISTSABIT P w = FORALLBITS ($~ o P) w

NOT_FORALLBITS (word bitop)
|- !P w. ~FORALLBITS P w = EXISTSABIT ($~ o P) w

PBITBOP_DEF (word bitop)
|- !op.

PBITBOP op =
!n (w1::PWORDLEN n) (w2::PWORDLEN n).

PWORDLEN n (op w1 w2) /\
!m k.
m + k <= n ==>
(op (WSEG m k w1) (WSEG m k w2) = WSEG m k (op w1 w2))

PBITBOP_EXISTS (word bitop)
|- !f. ?fn. !l1 l2. fn (WORD l1) (WORD l2) = WORD (MAP2 f l1 l2)

PBITBOP_PWORDLEN (word bitop)
|- !(op::PBITBOP) n (w1::PWORDLEN n) (w2::PWORDLEN n).

PWORDLEN n (op w1 w2)

PBITBOP_WSEG (word bitop)
|- !(op::PBITBOP) n (w1::PWORDLEN n) (w2::PWORDLEN n) m k.

m + k <= n ==>
(op (WSEG m k w1) (WSEG m k w2) = WSEG m k (op w1 w2))

PBITOP_BIT (word bitop)
|- !(op::PBITOP) n (w::PWORDLEN n) k.

k < n ==> (op (WORD [BIT k w]) = WORD [BIT k (op w)])

PBITOP_DEF (word bitop)
|- !op.

PBITOP op =
!n (w::PWORDLEN n).

PWORDLEN n (op w) /\
!m k. m + k <= n ==> (op (WSEG m k w) = WSEG m k (op w))

PBITOP_PWORDLEN (word bitop)
|- !(op::PBITOP) n (w::PWORDLEN n). PWORDLEN n (op w)

PBITOP_WSEG (word bitop)
|- !(op::PBITOP) n (w::PWORDLEN n) m k.

m + k <= n ==> (op (WSEG m k w) = WSEG m k (op w))

SHL_DEF (word bitop)
|- !f w b.

SHL f w b =
(BIT (PRE (WORDLEN w)) w,
WCAT

(WSEG (PRE (WORDLEN w)) 0 w,
(if f then WSEG 1 0 w else WORD [b])))

SHL_WSEG (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==>
0 < m ==>
!f b.

SHL f (WSEG m k w) b =
(BIT (k + (m - 1)) w,
(if f then

WCAT (WSEG (m - 1) k w,WSEG 1 k w)
else

WCAT (WSEG (m - 1) k w,WORD [b])))

SHL_WSEG_1F (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==>
0 < m ==>
!b.

SHL F (WSEG m k w) b =
(BIT (k + (m - 1)) w,WCAT (WSEG (m - 1) k w,WORD [b]))

SHL_WSEG_NF (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==>
0 < m ==>
0 < k ==>
(SHL F (WSEG m k w) (BIT (k - 1) w) =
(BIT (k + (m - 1)) w,WSEG m (k - 1) w))

SHR_DEF (word bitop)
|- !f b w.

SHR f b w =
(WCAT

((if f then WSEG 1 (PRE (WORDLEN w)) w else WORD [b]),
WSEG (PRE (WORDLEN w)) 1 w),BIT 0 w)

SHR_WSEG (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==>
0 < m ==>
!f b.

SHR f b (WSEG m k w) =
((if f then

WCAT (WSEG 1 (k + (m - 1)) w,WSEG (m - 1) (k + 1) w)
else

WCAT (WORD [b],WSEG (m - 1) (k + 1) w)),BIT k w)

SHR_WSEG_1F (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==>
0 < m ==>
!b.

SHR F b (WSEG m k w) =
(WCAT (WORD [b],WSEG (m - 1) (k + 1) w),BIT k w)

SHR_WSEG_NF (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k < n ==>
0 < m ==>
(SHR F (BIT (m + k) w) (WSEG m k w) = (WSEG m (k + 1) w,BIT k w))

WMAP_0 (word bitop)
|- !f. WMAP f (WORD []) = WORD []

WMAP_BIT (word bitop)
|- !n (w::PWORDLEN n) k. k < n ==> !f. BIT k (WMAP f w) = f (BIT k w)

WMAP_DEF (word bitop)
|- !f l. WMAP f (WORD l) = WORD (MAP f l)

WMAP_o (word bitop)
|- !w f g. WMAP g (WMAP f w) = WMAP (g o f) w

WMAP_PBITOP (word bitop)
|- !f. PBITOP (WMAP f)

WMAP_PWORDLEN (word bitop)
|- !(w::PWORDLEN n) f. PWORDLEN n (WMAP f w)

WMAP_WCAT (word bitop)
|- !w1 w2 f. WMAP f (WCAT (w1,w2)) = WCAT (WMAP f w1,WMAP f w2)

WMAP_WSEG (word bitop)
|- !n (w::PWORDLEN n) m k.

m + k <= n ==> !f. WMAP f (WSEG m k w) = WSEG m k (WMAP f w)

WSEG_SHL (word bitop)
|- !n (w::PWORDLEN (SUC n)) m k.

0 < k /\ m + k <= SUC n ==>
!b. WSEG m k (SND (SHL f w b)) = WSEG m (k - 1) w

WSEG_SHL_0 (word bitop)
|- !n (w::PWORDLEN (SUC n)) m b.

0 < m /\ m <= SUC n ==>
(WSEG m 0 (SND (SHL f w b)) =
WCAT (WSEG (m - 1) 0 w,(if f then WSEG 1 0 w else WORD [b])))

3.3 The theory word num

LVAL (word num)
|- (!f b. LVAL f b [] = 0) /\

!l f b x. LVAL f b (x::l) = f x * b EXP LENGTH l + LVAL f b l

LVAL_DEF (word num)
|- !f b l. LVAL f b l = FOLDL (\e x. b * e + f x) 0 l

LVAL_MAX (word num)
|- !l f b. (!x. f x < b) ==> LVAL f b l < b EXP LENGTH l

LVAL_SNOC (word num)
|- !l h f b. LVAL f b (SNOC h l) = LVAL f b l * b + f h

NLIST_DEF (word num)
|- (!frep b m. NLIST 0 frep b m = []) /\

!n frep b m.
NLIST (SUC n) frep b m =
SNOC (frep (m MOD b)) (NLIST n frep b (m DIV b))

NVAL0 (word num)
|- !f b. NVAL f b (WORD []) = 0

NVAL1 (word num)
|- !f b x. NVAL f b (WORD [x]) = f x

NVAL_DEF (word num)
|- !f b l. NVAL f b (WORD l) = LVAL f b l

NVAL_MAX (word num)
|- !f b. (!x. f x < b) ==> !n (w::PWORDLEN n). NVAL f b w < b EXP n

NVAL_WCAT (word num)
|- !n m (w1::PWORDLEN n) (w2::PWORDLEN m) f b.

NVAL f b (WCAT (w1,w2)) = NVAL f b w1 * b EXP m + NVAL f b w2

NVAL_WCAT1 (word num)
|- !w f b x. NVAL f b (WCAT (w,WORD [x])) = NVAL f b w * b + f x

NVAL_WCAT2 (word num)
|- !n (w::PWORDLEN n) f b x.

NVAL f b (WCAT (WORD [x],w)) = f x * b EXP n + NVAL f b w

NVAL_WORDLEN_0 (word num)
|- !(w::PWORDLEN 0) fv r. NVAL fv r w = 0

NWORD_DEF (word num)
|- !n frep b m. NWORD n frep b m = WORD (NLIST n frep b m)

NWORD_LENGTH (word num)
|- !n f b m. WORDLEN (NWORD n f b m) = n

NWORD_PWORDLEN (word num)
|- !n f b m. PWORDLEN n (NWORD n f b m)

3.4 The theory bword bitop

PBITBOP_WAND (bword bitop)
|- PBITBOP $WAND

PBITBOP_WOR (bword bitop)
|- PBITBOP $WOR

PBITBOP_WXOR (bword bitop)
|- PBITBOP $WXOR

PBITOP_WNOT (bword bitop)
|- PBITOP WNOT

WAND_DEF (bword bitop)
|- !l1 l2. WORD l1 WAND WORD l2 = WORD (MAP2 $/\ l1 l2)

WCAT_WNOT (bword bitop)
|- !n1 n2 (w1::PWORDLEN n1) (w2::PWORDLEN n2).

WCAT (WNOT w1,WNOT w2) = WNOT (WCAT (w1,w2))

WNOT_DEF (bword bitop)
|- !l. WNOT (WORD l) = WORD (MAP $~ l)

WNOT_WNOT (bword bitop)
|- !w. WNOT (WNOT w) = w

WOR_DEF (bword bitop)
|- !l1 l2. WORD l1 WOR WORD l2 = WORD (MAP2 $\/ l1 l2)

WXOR_DEF (bword bitop)
|- !l1 l2. WORD l1 WXOR WORD l2 = WORD (MAP2 (\x y. ~(x = y)) l1 l2)

3.5 The theory bword num

ADD_BNVAL_LEFT (bword num)
|- !n (w1::PWORDLEN (SUC n)) (w2::PWORDLEN (SUC n)).

BNVAL w1 + BNVAL w2 =
(BV (BIT n w1) + BV (BIT n w2)) * 2 EXP n +
(BNVAL (WSEG n 0 w1) + BNVAL (WSEG n 0 w2))

ADD_BNVAL_RIGHT (bword num)
|- !n (w1::PWORDLEN (SUC n)) (w2::PWORDLEN (SUC n)).

BNVAL w1 + BNVAL w2 =
(BNVAL (WSEG n 1 w1) + BNVAL (WSEG n 1 w2)) * 2 +
(BV (BIT 0 w1) + BV (BIT 0 w2))

ADD_BNVAL_SPLIT (bword num)
|- !n1 n2 (w1::PWORDLEN (n1 + n2)) (w2::PWORDLEN (n1 + n2)).

BNVAL w1 + BNVAL w2 =
(BNVAL (WSEG n1 n2 w1) + BNVAL (WSEG n1 n2 w2)) * 2 EXP n2 +
(BNVAL (WSEG n2 0 w1) + BNVAL (WSEG n2 0 w2))

BIT_NBWORD0 (bword num)
|- !k n. k < n ==> (BIT k (NBWORD n 0) = F)

BNVAL0 (bword num)
|- BNVAL (WORD []) = 0

BNVAL_11 (bword num)
|- !w1 w2.

(WORDLEN w1 = WORDLEN w2) ==> (BNVAL w1 = BNVAL w2) ==> (w1 = w2)

BNVAL_DEF (bword num)
|- !l. BNVAL (WORD l) = LVAL BV 2 l

BNVAL_MAX (bword num)
|- !n (w::PWORDLEN n). BNVAL w < 2 EXP n

BNVAL_NBWORD (bword num)
|- !n m. m < 2 EXP n ==> (BNVAL (NBWORD n m) = m)

BNVAL_NVAL (bword num)
|- !w. BNVAL w = NVAL BV 2 w

BNVAL_ONTO (bword num)
|- !w. ?n. BNVAL w = n

BNVAL_WCAT (bword num)
|- !n m (w1::PWORDLEN n) (w2::PWORDLEN m).

BNVAL (WCAT (w1,w2)) = BNVAL w1 * 2 EXP m + BNVAL w2

BNVAL_WCAT1 (bword num)
|- !n (w::PWORDLEN n) x. BNVAL (WCAT (w,WORD [x])) = BNVAL w * 2 + BV x

BNVAL_WCAT2 (bword num)
|- !n (w::PWORDLEN n) x.

BNVAL (WCAT (WORD [x],w)) = BV x * 2 EXP n + BNVAL w

BV_DEF (bword num)
|- !b. BV b = (if b then SUC 0 else 0)

BV_LESS_2 (bword num)
|- !x. BV x < 2

BV_VB (bword num)
|- !x. x < 2 ==> (BV (VB x) = x)

DOUBL_EQ_SHL (bword num)
|- !n.

0 < n ==>
!(w::PWORDLEN n) b.

NBWORD n (BNVAL w + BNVAL w + BV b) = SND (SHL F w b)

EQ_NBWORD0_SPLIT (bword num)
|- !n (w::PWORDLEN n) m.

m <= n ==>
((w = NBWORD n 0) =
(WSEG (n - m) m w = NBWORD (n - m) 0) /\
(WSEG m 0 w = NBWORD m 0))

MSB_NBWORD (bword num)
|- !n m. BIT n (NBWORD (SUC n) m) = VB ((m DIV 2 EXP n) MOD 2)

NBWORD0 (bword num)
|- !m. NBWORD 0 m = WORD []

NBWORD_BNVAL (bword num)
|- !n (w::PWORDLEN n). NBWORD n (BNVAL w) = w

NBWORD_DEF (bword num)
|- !n m. NBWORD n m = WORD (NLIST n VB 2 m)

NBWORD_MOD (bword num)
|- !n m. NBWORD n (m MOD 2 EXP n) = NBWORD n m

NBWORD_SPLIT (bword num)
|- !n1 n2 m.

NBWORD (n1 + n2) m = WCAT (NBWORD n1 (m DIV 2 EXP n2),NBWORD n2 m)

NBWORD_SUC (bword num)
|- !n m.

NBWORD (SUC n) m = WCAT (NBWORD n (m DIV 2),WORD [VB (m MOD 2)])

NBWORD_SUC_FST (bword num)
|- !n m.

NBWORD (SUC n) m =
WCAT (WORD [VB ((m DIV 2 EXP n) MOD 2)],NBWORD n m)

NBWORD_SUC_WSEG (bword num)
|- !n (w::PWORDLEN (SUC n)). NBWORD n (BNVAL w) = WSEG n 0 w

PWORDLEN_NBWORD (bword num)
|- !n m. PWORDLEN n (NBWORD n m)

VB_BV (bword num)
|- !x. VB (BV x) = x

VB_DEF (bword num)
|- !n. VB n = ~(n MOD 2 = 0)

WCAT_NBWORD_0 (bword num)
|- !n1 n2. WCAT (NBWORD n1 0,NBWORD n2 0) = NBWORD (n1 + n2) 0

WORDLEN_NBWORD (bword num)
|- !n m. WORDLEN (NBWORD n m) = n

WSEG_NBWORD (bword num)
|- !m k n.

m + k <= n ==> !l. WSEG m k (NBWORD n l) = NBWORD m (l DIV 2 EXP k)

WSEG_NBWORD_SUC (bword num)
|- !n m. WSEG n 0 (NBWORD (SUC n) m) = NBWORD n m

WSPLIT_NBWORD_0 (bword num)
|- !n m.

m <= n ==> (WSPLIT m (NBWORD n 0) = (NBWORD (n - m) 0,NBWORD m 0))

ZERO_WORD_VAL (bword num)
|- !n (w::PWORDLEN n). (w = NBWORD n 0) = (BNVAL w = 0)

3.6 The theory bword arith

ACARRY_ACARRY_WSEG (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) cin m k1 k2.

k1 < m /\ k2 < n /\ m + k2 <= n ==>
(ACARRY k1 (WSEG m k2 w1) (WSEG m k2 w2) (ACARRY k2 w1 w2 cin) =
ACARRY (k1 + k2) w1 w2 cin)

ACARRY_DEF (bword arith)
|- (!w1 w2 cin. ACARRY 0 w1 w2 cin = cin) /\

!n w1 w2 cin.
ACARRY (SUC n) w1 w2 cin =
VB

((BV (BIT n w1) + BV (BIT n w2) + BV (ACARRY n w1 w2 cin)) DIV 2)

ACARRY_EQ_ADD_DIV (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) k.

k < n ==>
(BV (ACARRY k w1 w2 cin) =
(BNVAL (WSEG k 0 w1) + BNVAL (WSEG k 0 w2) + BV cin) DIV 2 EXP k)

ACARRY_EQ_ICARRY (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) cin k.

k <= n ==> (ACARRY k w1 w2 cin = ICARRY k w1 w2 cin)

ACARRY_MSB (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) cin.

ACARRY n w1 w2 cin =
BIT n (NBWORD (SUC n) (BNVAL w1 + BNVAL w2 + BV cin))

ACARRY_WSEG (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) cin k m.

k < m /\ m <= n ==>
(ACARRY k (WSEG m 0 w1) (WSEG m 0 w2) cin = ACARRY k w1 w2 cin)

ADD_NBWORD_EQ0_SPLIT (bword arith)
|- !n1 n2 (w1::PWORDLEN (n1 + n2)) (w2::PWORDLEN (n1 + n2)) cin.

(NBWORD (n1 + n2) (BNVAL w1 + BNVAL w2 + BV cin) =
NBWORD (n1 + n2) 0) =
(NBWORD n1

(BNVAL (WSEG n1 n2 w1) + BNVAL (WSEG n1 n2 w2) +
BV (ACARRY n2 w1 w2 cin)) =

NBWORD n1 0) /\
(NBWORD n2 (BNVAL (WSEG n2 0 w1) + BNVAL (WSEG n2 0 w2) + BV cin) =
NBWORD n2 0)

ADD_WORD_SPLIT (bword arith)
|- !n1 n2 (w1::PWORDLEN (n1 + n2)) (w2::PWORDLEN (n1 + n2)) cin.

NBWORD (n1 + n2) (BNVAL w1 + BNVAL w2 + BV cin) =
WCAT

(NBWORD n1
(BNVAL (WSEG n1 n2 w1) + BNVAL (WSEG n1 n2 w2) +
BV (ACARRY n2 w1 w2 cin)),

NBWORD n2
(BNVAL (WSEG n2 0 w1) + BNVAL (WSEG n2 0 w2) + BV cin))

ICARRY_DEF (bword arith)
|- (!w1 w2 cin. ICARRY 0 w1 w2 cin = cin) /\

!n w1 w2 cin.
ICARRY (SUC n) w1 w2 cin =
BIT n w1 /\ BIT n w2 \/
(BIT n w1 \/ BIT n w2) /\ ICARRY n w1 w2 cin

ICARRY_WSEG (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) cin k m.

k < m /\ m <= n ==>
(ICARRY k (WSEG m 0 w1) (WSEG m 0 w2) cin = ICARRY k w1 w2 cin)

WSEG_NBWORD_ADD (bword arith)
|- !n (w1::PWORDLEN n) (w2::PWORDLEN n) m k cin.

m + k <= n ==>
(WSEG m k (NBWORD n (BNVAL w1 + BNVAL w2 + BV cin)) =
NBWORD m
(BNVAL (WSEG m k w1) + BNVAL (WSEG m k w2) +
BV (ACARRY k w1 w2 cin)))

